
Permutation is an operation for matching that shall be used only on values of record of and array types. Permutation is denoted by the keyword permutation. Expressions, templates and AnyElement and AnyElementsOrNone are allowed as permutation elements. Permutation elements shall obey the restrictions given below.
A permutation without AnyElementsOrNone in place of a single record of element means that any series of elements is acceptable provided that there is a one to one mapping between elements in the record of and in the permutation list such that each element matches its corresponding element in the permutation list. 
AnyElementsOrNone used inside permutation (directly or via reference) replaces none or any number of elements within the segment of the record of value matched by permutation. The permutation matching is successful, if a subset of the elements in the record of matches the permutation list without the AnyElementsOrNone. If both permutation and AnyElementsOrNone are used in a record of template, they shall be evaluated jointly.
NOTE 1:	AnyElementsOrNone used inside permutation has a different effect as AnyElementsOrNone used in conjunction with permutation as in the latter AnyElementsOrNone replaces consecutive elements only. For example, {permutation(1,2,*)} is equivalent to ({*,1,*,2,*},{*,2,*,1,*}), while {permutation(1,2),*} is equivalent to ({1,2,*},{2,1,*}).
NOTE 2:	When AnyElementsOrNone is inside a permutation, a length attribute may be applied to AnyElementsOrNone to restrict the number of elements matched by AnyElementsOrNone (see also clause B.1.4.1). 
Besides specifying all individual values, it is possible to add all elements of a record of or set of template into permutations using an all from clause.
Restrictions
a)	Each individual member listed in the permutation shall be of the type replicated by the record of or array type.
b)	The member type of the permutation and the member type of the template in the all from clause shall be compatible.
[bookmark: annex_Matching_Permutation_ResTemplCont]c)	The template in the all from clause as a whole shall not resolve into a matching mechanism other than a SpecificValue (see clause B.1.1), and its elements may resolve to the matching mechanisms SpecificValue, AnyElement and AnyElementsOrNone only.
[bookmark: bugnotes][bookmark: _GoBack]d)	Should individual template members of the permutation resolve to an all from template clause, the template in the all from clause shall obeying to restriction c) above.Individual members of a permutation and elements of the template in the all from clause shall only be expressions, templates obeying to restriction c) above, and the AnyElement and AnyElementsOrNone matching mechanisms.
Examples
EXAMPLE 1:
	type record of integer MySequenceOfType;

	template MySequenceOfType mw_myTemplate1 := { permutation ( 1, 2, 3 ), 5 };
	// matches any of the following sequences of 4 integers: 1,2,3,5; 1,3,2,5; 2,1,3,5;
	// 2,3,1,5; 3,1,2,5; or 3,2,1,5

	template MySequenceOfType mw_myTemplate2 := { permutation ( 1, 2, ? ), 5 };
	// matches any sequence of 4 integers that ends with 5 and contains 1 and 2 at least once in
	// other positions

	template MySequenceOfType mw_myTemplate3 := { permutation ( 1, 2, 3 ), * };
	// matches any sequence of integers starting with 1,2,3; 1,3,2; 2,1,3; 2,3,1; 3,1,2 or 3,2,1

	template MySequenceOfType mw_myTemplate4 := { *, permutation ( 1, 2, 3 )};
	// matches any sequence of integers ending with 1,2,3; 1,3,2; 2,1,3; 2,3,1; 3,1,2 or 3,2,1

	template MySequenceOfType mw_myTemplate5 := { *, permutation ( 1, 2, 3 ),* };
	// matches any sequence of integers containing any of the following substrings at any position:
	// 1,2,3; 1,3,2; 2,1,3; 2,3,1; 3,1,2 or 3,2,1 

	template MySequenceOfType mw_myTemplate6 := { permutation ( 1, 2, * ), 5 };
	// matches any sequence of integers that ends with 5 and containing 1 and 2 at least once in
	// other positions

	template MySequenceOfType mw_myTemplate7 := { permutation ( 1, 2, 3 ), * length (0..5)};
	// matches any sequence of three to eight integers starting with 1,2,3; 1,3,2; 2,1,3; 2,3,1;
	// 3,1,2 or 3,2,1

	template integer mw_myInt1 := (1,2,3);
	template integer mw_myInt2 := (1,2,?);
	template integer mw_myInt3 := ?;
	template integer mw_myInt4 := *;

	template MySequenceOfType mw_myTemplate10 := { permutation (mw_myInt1, 2, 3 ), 5 };
	// matches any of the sequences of 4 integers:
	//		1,3,2,5; 2,1,3,5; 2,3,1,5; 3,1,2,5; or 3,2,1,5;
	//		2,3,2,5; 2,2,3,5; 2,3,2,5; 3,2,2,5; or 3,2,2,5;
	//		3,3,2,5; 2,3,3,5; 2,3,3,5; 3,3,2,5; or 3,2,3,5;

	template MySequenceOfType mw_myTemplate11 := { permutation (mw_myInt2, 2, 3 ), 5 };
	// matches any sequence of 4 integers that ends with 5 and contains 2 and 3 at least once in
	// other positions

	template MySequenceOfType mw_myTemplate12 := { permutation (mw_myInt3, 2, 3 ), 5 };
	// matches any sequence of 4 integers that ends with 5 and contains 2 and 3 at least once in
	// other positions

	template MySequenceOfType mw_myTemplate13 := { permutation (mw_myInt4, 2, 3 ), 5 };
	// matches any sequence of integers that ends with 5 and containing 2 and 3 at least once in
	// other positions

	template MySequenceOfType mw_myTemplate14 := { permutation (mw_myInt3, 2, ? ), 5 };
	// matches any sequence of 4 integers that ends with 5 and contains 2 at least once in
	// other positions

	template MySequenceOfType mw_myTemplate15 := { permutation (mw_myInt4, 2, * ), 5 };
	// matches any sequence of integers that ends with 5 and contains 2 at least once in
	// other positions

EXAMPLE 2:
	type record of integer RoI;
	template RoI mw_roI1 := {1, 2, *};

	template RoI mw_roI2 := {permutation(0, all from mw_roI1), 4, 5}; 
	// results in {permutation(0, 1, 2, *), 4, 5}

